
PROGRAM TIL INVERSION AF LAPLACE TRANSFORM 120365 edda sveinsdottir
dog først
Procedurer til styring af CALCOMP−plotter.
_b_e_g_i_n
_i_n_t_e_g_e_r xxx,yyy,l111,pl111,plabc; _r_e_a_l deltax,deltay;

_p_r_o_c_e_d_u_r_e plotgraph(x0,x1,x,y,dx);
_v_a_l_u_e x0,x1; _r_e_a_l x0,x1,x,y,dx;
_b_e_g_i_n _i_n_t_e_g_e_r xx,yy,tr; _b_o_o_l_e_a_n hop; _b_o_o_l_e_a_n _a_r_r_a_y B[1:l111];
tr:=drumplace; drumplace:=pl111;
from drum(B); drumplace:=tr;
x:=x0; xx:=x0/deltax−xxx; yy:=y/deltay−yyy;
gierproc(B[2],hop,xx,yy);
xxx:=xxx+xx; yyy:=yyy+yy;
_f_o_r x:=x0 _s_t_e_p dx _u_n_t_i_l x1 _d_o
_b_e_g_i_n
     xx:=x/deltax−xxx; yy:=y/deltay−yyy;
     gier(hop);
     xxx:=xxx+xx; yyy:=yyy+yy
_e_n_d
_e_n_d plotgraph;

_p_r_o_c_e_d_u_r_e plotline(x0,y0,x1,y1);
_v_a_l_u_e x0,y0,x1,y1; _r_e_a_l x0,y0,x1,y1;
_b_e_g_i_n _i_n_t_e_g_e_r xx,yy,tr; _b_o_o_l_e_a_n hop; _b_o_o_l_e_a_n _a_r_r_a_y B[1:l111];
tr:=drumplace; drumplace:=pl111;
from drum(B); drumplace:=tr;
xx:=x0/deltax−xxx; yy:=y0/deltay−yyy;
gierproc(B[2],hop,xx,yy);
xxx:=xxx+xx; yyy:=yyy+yy;
xx:=x1/deltax−xxx; yy:=y1/deltay−yyy;
gier(hop);
xxx:=xxx+xx; yyy:=yyy+yy
_e_n_d plotline;

_r_e_a_l _p_r_o_c_e_d_u_r_e plotchar(t,x,y,h);
_v_a_l_u_e t,x,y,h; _i_n_t_e_g_e_r t; _r_e_a_l x,y,h;
_b_e_g_i_n _i_n_t_e_g_e_r xx,yy,n1,n2,hh,i,x1,y1; _b_o_o_l_e_a_n hop,AA,flyt;
_b_o_o_l_e_a_n _a_r_r_a_y A[0:1],B[1:l111];
_i_f t_>58 _t_h_e_n _g_o_t_o exit; hh:=h×20;
xx:=drumplace; drumplace:=pl111;
from drum(B); drumplace:=plabc−(57−t)×2;
from drum(A); drumplace:=xx;
AA:=A[0];
x1:=split(AA,0,4,n1,5,9,y1)_:6;
y1:=(y1−x1×6)×hh; x1:=x1×hh;
xx:=x1+x/deltax−xxx; yy:=y1+y/deltay−yyy;
xxx:=x/deltax; yyy:=y/deltay;
_i_f kbon _t_h_e_n write(|<−nddd|>,writecr,xx,yy);
gierproc(B[2],hop,xx,yy);
n2:=(n1−8)×5; n1:=_i_f n1_<7 _t_h_e_n n1×5 _e_l_s_e 35;
flyt:=_f_a_l_s_e;
_f_o_r i:=10 _s_t_e_p 5 _u_n_t_i_l n1,0 _s_t_e_p 5 _u_n_t_i_l n2 _d_o
_b_e_g_i_n
     _i_f i=0 _t_h_e_n AA:=A[1];
     xx:=split(AA,i,i+4,yy)_:6;
     _i_f yy=31 _t_h_e_n
     _b_e_g_i_n
        flyt:=_t_r_u_e;
     _e_n_d
     _e_l_s_e
     _b_e_g_i_n
     yy:=(yy−xx×6)×hh−y1;



     xx:=xx×hh−x1;
_i_f kbon _t_h_e_n write(|<−nddd|>,writecr,xx,yy);
_i_f flyt _t_h_e_n
_b_e_g_i_n
   flyt:=_f_a_l_s_e;
gierproc(B[2],hop,xx,yy);
_e_n_d _e_l_s_e
     gier(hop);
     x1:=x1+xx; y1:=y1+yy
     _e_n_d
     _e_n_d i;
xxx:=xxx+x1; yyy:=yyy+y1;
exit:plotchar:=_i_f t<58 _t_h_e_n x+100×h×deltax _e_l_s_e x _e_n_d plotchar;
xxx:=yyy:=0; deltax:=deltay:=0.01;
pl111:=drumplace; gierdrum(111,l111);
plabc:=drumplace; gierdrum(113,0);

_c_o_m_m_e_n_t program af 120365 til inversion af Laplace transform = f(t)
arbejder med 3 slags data:
      1)  syntetiske data af form f(t)=Sum[N[i]×exp(−lambda[i]×t)]
som det selv genererer ved hjælp af _r_e_a_l _p_r_o_c_e_d_u_r_e f(x). I dette
tilfælde skrives o på skrivemaskinen inden kompileringen, for at
inkludere procedurekrop svarende til denne type data. pause−
signalet ignoreres.
      2)  syntetiske data, som ikke er en sum af eksponential
funktioner, men hvor f(t)=f(exp(x)) kan udtrykkes på explicit form.
I dette tilfælde skrives der _i_k_k_e o på skrivemaskinen, men når
pause−signalet kommer, tastes procedurekroppen til _r_e_a_l _p_r_o_c_e_d_u_r_e
f(x) ind fra skrivemaskinen. Eks:
          f:= exp(−0.001×exp(x))/(exp(x)+0.009)|∧ 2.
      3)  experimentelle data i form af et array af funktionsværdier
aflæst for ækvidistante værdier af lnt=x. I dette tilfælde tastes o
før compileringen. Ved at gøre n negativ (se nedenfor) vil maskinen
være parat til at indlæse en strimmel indeholdende data−array F[n,−n].
pause−signalet ignoreres.
DATASTRIMMEL I: maskinprocedure plot 111, 112, 113.
DATASTRIMMEL II:bestaar af tabel over den komplekse funktion,
Gamma(1+iu) hvor (u=0(0.1)10).
INPUT FRA SKRIVEMASKINE:
dx     skridtlængde ved x−integration  (f.eks. 0.25)
du     skridtlængde ved u−integration, skal være mult af 0.1
xmax   øvre grænse ved x−integrationen, tastes evt først ind , efter
       at plotteren har tegnet grafen af integranden ved x−integration
       skal være lige mult af dx
umax   øvre grænse ved u−integration, skal være lige mult af du
lamax  hhv øvre og nedre grænse for det interval for lamda, hvori
lamin  man ønsker at undersøge g(lamda)/lamda. lamax og lamin skal
       hele mult af 10−i
noofy  antal plottepunkter i det angivne lamda−interval
n, n>0 antal exponentialled i den syntetiske kurve f(t) af typen 1)
       med parametrene
N[i], lambda[i]   (i=1,...,n)
n, n<0 i dette tilfælde indlæses strimmel med rigtige data
F[n,−n] som er den experimentelle kurve aflæst med ækvidistancen dx
       i x−intervallet [n×dx,−n×dx];

_b_e_g_i_n _i_n_t_e_g_e_r n,m,noofy,Kplads;  _b_o_o_l_e_a_n nykurve;
_r_e_a_l dx,xmax,du,umax,lamax,lamin;  _a_r_r_a_y Kc,Ks[0:100];

_p_r_o_c_e_d_u_r_e plotcond(x,y,t,e1,e2,b1,b2);
_r_e_a_l x,y,t,e1,e2; _b_o_o_l_e_a_n b1,b2;
_b_e_g_i_n _i_n_t_e_g_e_r tr,xx,yy; _b_o_o_l_e_a_n hop,s;
_b_o_o_l_e_a_n _a_r_r_a_y B[1:l111];
tr:= drumplace; drumplace:= pl111;



from drum(B); drumplace:= tr; s:= _t_r_u_e;
_f_o_r t:= e1,e2 _w_h_i_l_e b1 _d_o
_i_f −, b2 _t_h_e_n s:= _t_r_u_e _e_l_s_e _b_e_g_i_n
xx:= x/deltax − xxx; yy:= y/deltay − yyy;
_i_f s _t_h_e_n _b_e_g_i_n gierproc(B[2],hop,xx,yy);
xxx:= xxx+xx; yyy:= yyy+yy;
s:= _f_a_l_s_e _e_n_d _e_l_s_e _b_e_g_i_n gier(hop);
xxx:= xxx+xx; yyy:= yyy+yy _e_n_d
_e_n_d
_e_n_d plotcond;

_f_o_r n:=0 _s_t_e_p 1 _u_n_t_i_l 100 _d_o input(Kc[n], Ks[n]);  n:= 0;
DATA:
writecr;  writetext(|<<dx,xmax,du,umax,lamax,lamin,noofy:
|>);  outcr;  outtext(|<<program 120365 − inversion af Laplace transform|>);
dx:= typein;  xmax:= typein;  du:= typein;  umax:= typein;
lamax:= typein;  lamin:= typein;  noofy:= typein; m:= 1;
writecr;  writetext(|<<m:|>); m:= typein;
_i_f m=0 _t_h_e_n _g_o_t_o NÆSTYDERSTE BLOK;
_i_f m<0 _t_h_e_n _b_e_g_i_n n:= −m;  nykurve:= _t_r_u_e; _g_o_t_o NÆSTYDERSTE BLOK _e_n_d;
n:= m;  writecr;  writetext(|<<N(i), lambda(i): |>);  nykurve:= _t_r_u_e;

NÆSTYDERSTE BLOK:
_b_e_g_i_n _i_n_t_e_g_e_r i,j,k,uo,us,K,S,grænse;
_r_e_a_l x,u,y,ymax,ymin,dy,gy,pi,Fcu,Fsu,Kcu,Ksu,nævner,lamda,lstep,gmax;
_a_r_r_a_y N,lambda[1:n],g[0:noofy],F[−n:n];
_r_e_a_l _p_r_o_c_e_d_u_r_e f(x);  _v_a_l_u_e x;  _r_e_a_l x;
_b_e_g_i_n _r_e_a_l fx;  fx:= 0;
_f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o fx:= fx + N[i]×exp(−lambda[i]×exp(x));
f:= fx;
_e_n_d f;*;
_i_f m<0 _t_h_e_n _b_e_g_i_n input(F);  outcr;  outtext(|<<real data|>);  _g_o_t_o XINT _e_n_d;
_i_f m>0 _t_h_e_n _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o
  _b_e_g_i_n N[i]:= typein;  lambda[i]:= typein _e_n_d;
outcr;  outcr;  outtext(|<<N(i), lambda(i): |>);
_f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l n _d_o output(|<−ndddd.ddd|>,outcr,N[i],lambda[i]);
_i_f m=0 _t_h_e_n _g_o_t_o XINT;

_b_e_g_i_n _r_e_a_l DX;
_c_o_m_m_e_n_t i denne blok bliver log y, hvor y=exp(x)×f(x), plottet mod x
(x=−10,−9,...,10) i skridt af o.25. x−enheden=0.25 tomme. y−dekaden
=1.5 tomme. Der er 5 dekader;
deltax:= 0.04;  deltay:= 0.02/3×2.302585;
deltax:= 0.02;  deltay:= 0.01/3×2.302585;
_c_o_m_m_e_n_t faktoren 2.302585 i deltay er lig ln 10. Dvs kan ved kald af
plotprocedurer benytte ln i stedet for log10;  writetext(|<<
GØR CALCOMP KLAR t×f(t) plottes mod ln(t) − hvis dette uønsket tast 0: |>);
j:= typein;  _i_f j=0 _t_h_e_n _g_o_t_o XINT;
x:= −10;  DX:= 0.4; xxx:= yyy:= 0;
YAXE:
_f_o_r j:=1,10,100,1000,10000 _d_o
  _f_o_r k:=2 _s_t_e_p 1 _u_n_t_i_l 10 _d_o
  _b_e_g_i_n plotline(x,ln((k−1)×j),x,ln(k×j));
    plotline(x,ln(k×j),x+DX,ln(k×j));
  _e_n_d k;  _i_f x=10 _t_h_e_n _g_o_t_o XAXE;
_f_o_r k:=6 _s_t_e_p −1 _u_n_t_i_l 1 _d_o
_b_e_g_i_n x:= −10−k×0.7;  j:= 1;
  _f_o_r j:=j _w_h_i_l_e x<−10.6 _d_o
  _b_e_g_i_n x:= plotchar(j,x,ln(10|∧ (k−1)),0.175);  j:= 16 _e_n_d;
_e_n_d k;
x:= 10; DX:= −0.4;  _g_o_t_o YAXE;
XAXE:
deltay:= deltay/2.302585×1.5;  _c_o_m_m_e_n_t y−enhed = 1 tomme;
_f_o_r j:=8 _s_t_e_p −2 _u_n_t_i_l −10 _d_o



_b_e_g_i_n plotline(j+2,0,j,0);  plotline(j,0,j,−0.1);
  _i_f j=0 _t_h_e_n
  _b_e_g_i_n plotchar(16,−0.24,−0.4,0.15);  x:= plotchar(23,−1.5,−0.9,0.15);
    plotline(x+0.16,−0.8,x+0.44,−0.8);  plotline(x+0.16,−0.85,x+0.44,−0.85);
    x:= x+0.6;  _f_o_r i:=35,37,19 _d_o x:= plotchar(i,x,−0.9,0.15);
  _e_n_d _e_l_s_e
    _i_f j>0 _t_h_e_n plotchar(j,j−0.24,−0.4,0.15) _e_l_s_e
      _i_f j|=−10 _t_h_e_n _b_e_g_i_n plotline(j−0.8,−0.325,j−0.55,−0.325);
        plotchar(−j,j−0.24,−0.4,0.15) _e_n_d;
_e_n_d;
  x:= −8;  x:= plotchar(53,x,7,0.15);  x:= plotchar(23,x,7.1,0.1);
  _f_o_r j:=54,53 _d_o x:= plotchar(j,x,7,0.15);  plotchar(23,x,7.1,0.1);
deltay:= deltay×2.302585/1.5; _c_o_m_m_e_n_t y−dekade = 1.5 tomme;
plotcond(x,x+ln(f(x)),x,−10,x+0.15,x_<10,x+ln(f(x))>ln(0.5));
writecr;  writetext(|<<xmax (lige mult af dx): |>);  xmax:= typein;
_e_n_d paa blok med plotning af exp(x)×f(x);

XINT:  grænse:= entier(xmax/dx);  j:= entier(umax×10);
_b_e_g_i_n _r_e_a_l k1,k2,k3,k4,k5;  _a_r_r_a_y Fc,Fs[0:j];  _i_n_t_e_g_e_r u1,du1,umax1;
_b_e_g_i_n _r_e_a_l plus,minus,Fcos,Fsin;  _a_r_r_a_y fc,fs[0:grænse];
_c_o_m_m_e_n_t i denne blok foretages x−integrationen ved hjælp af
Simpson. Dvs Fc(u) og Fs(u), hhv den reelle og imaginære del
af F(u) beregnes;
_f_o_r k:=0 _s_t_e_p 1 _u_n_t_i_l grænse _d_o
_b_e_g_i_n x:= k×dx;
   write(|<nddd|>,writecr,writetext(|<<k,grænse I: |>),k,grænse);
  _i_f m<0
    _t_h_e_n _b_e_g_i_n plus:= exp(x)×F[k];  minus:= exp(−x)×F[−k] _e_n_d
    _e_l_s_e _b_e_g_i_n plus:= exp(x)×f(k);  minus:= exp(−x)×f(−k) _e_n_d;
  fc[k]:= plus+minus;   fs[k]:= plus−minus;
_e_n_d k;
du1:= du×10;  umax1:= umax×10;
   write(|<nddd.dd|>,writecr,writetext(|<<du,du1: |>),du,du1);
_f_o_r u1:=0 _s_t_e_p du1 _u_n_t_i_l umax1 _d_o
_b_e_g_i_n u:= u1/10;
  writecr; writetext(|<<kilroy 1|>);
  Fcos:= fc[0]+4×fc[1]×cos(u×dx)+fc[grænse]×cos(u×xmax);
  Fsin:= 4×fs[1]×sin(u×dx)+fs[grænse]×sin(u×xmax);
  _f_o_r k:=2 _s_t_e_p  2 _u_n_t_i_l grænse−2 _d_o
  _b_e_g_i_n
   write(|<nddd|>,writecr,writetext(|<<k,grænse II: |>),k,grænse);
    Fcos:= Fcos + 2×fc[k]×cos(u×k×dx) + 4×fc[k+1]×cos(u×(k+1)×dx);
    Fsin:= Fsin + 2×fs[k]×sin(u×k×dx) + 4×fs[k+1]×sin(u×(k+1)×dx);
  _e_n_d k;
   write(|<nddd|>,writecr,writetext(|<<u1,du1,umax1: |>),u1,du1,umax1);
  Fc[u1] := Fcos×dx/3;
  writecr; writetext(|<<kilroy 2|>);
  Fs[u1] := Fsin×dx/3;
  writecr; writetext(|<<kilroy 3|>);
_e_n_d u;
_e_n_d x−integration;

_c_o_m_m_e_n_t nu kendes Fc,Fs,Kc,Ks og u−integrationen kan
foretages (Simpson);
uo:= entier(umax/0.1);  us:= du/0.1;
write(|<ndddd|>,writecr,writetext(|<<uo,us: |>),uo,us);
ymax:= −ln(lamin);  ymin:= −ln(lamax);
dy:= (ymax−ymin)/noofy;  i:= 0;
writecr; writetext(|<<k1:|>);
k1:= (Fc[0]×Kc[0]+Fs[0]×Ks[0])/(Kc[0]|∧ 2+Ks[0]|∧ 2);
writecr; writetext(|<<k2:|>);
k2:= 4×(Fc[us]×Kc[us]+Fs[us]×Ks[us])/(Kc[us]|∧ 2+Ks[us]|∧ 2);
writecr; writetext(|<<k3:|>);
k3:= 4×(Fs[us]×Kc[us]−Fc[us]×Ks[us])/(Kc[us]|∧ 2+Ks[us]|∧ 2);



writecr; writetext(|<<k4:|>);
k4:= (Fc[uo]×Kc[uo]+Fs[uo]×Ks[uo])/(Kc[uo]|∧ 2+Ks[uo]|∧ 2);
writecr; writetext(|<<k5:|>);
k5:= (Fs[uo]×Kc[uo]−Fc[uo]×Ks[uo])/(Kc[uo]|∧ 2+Ks[uo]|∧ 2);
_f_o_r y:=ymin _s_t_e_p dy _u_n_t_i_l ymax _d_o
_b_e_g_i_n
  gy:= k1+k2×cos(y×du)+k3×sin(y×du)+k4×cos(y×umax)+k5×sin(y×umax);
  _f_o_r k:=2×us _s_t_e_p 2×us _u_n_t_i_l uo−2 _d_o
  _b_e_g_i_n K:= k;   S:= 2;
OM: Fcu:= Fc[K];  Fsu:= Fs[K];  Kcu:= Kc[K];  Ksu:= Ks[K];
    nævner:= Kcu|∧ 2 + Ksu|∧ 2;  u:= K×0.1;
    gy:= gy+S×(Fcu×Kcu+Fsu×Ksu)/nævner×cos(y×u)
           +S×(Fsu×Kcu−Fcu×Ksu)/nævner×sin(y×u);
    K:=K + us;  S:= 4;  _i_f K= k+us _t_h_e_n _g_o_t_o OM;
  _e_n_d k;
  g[i]:= gy;  i:= i+1;
_e_n_d y og u−integration;
_e_n_d integrationer;

_c_o_m_m_e_n_t nu udskrives og plottes resultaterne;
outcr;  outcr;  outtext(|<<dx,xmax,du,umax,lamax,lamin,noofy: |>);
outcr;  output(|<−nddd.dddd|>, dx,xmax,du,umax,lamax,lamin,noofy);
outcr;  outcr;  outtext(|<<y,lambda,g(exp(−y)): |>);
_f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l noofy _d_o
_b_e_g_i_n y:= ymin + i×dy;  lamda:= exp(−y);  outcr;
  output(|<−ndd.dd|>, y);  output(|<−nddd.dddd|>, lamda);
  output(|<−ndd.d10+d|>, outsp(4), g[i]);
_e_n_d i;
gmax:= g[0];
_f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l noofy _d_o _i_f gmax<g[i] _t_h_e_n gmax:= g[i];
outcr;  outcr;  outtext(|<<100 = |>);  output(|<−ndd.d10+d|>, gmax);
_f_o_r j:=0 _s_t_e_p 1 _u_n_t_i_l noofy _d_o g[j]:= g[j]/gmax×100;
writecr;  writetext(|<<gør Calcomp klar til ny plotning|>);  j:= typein;
deltax:= 0.01×(ymax−ymin)/5;
deltay:= 0.01×100/5; _c_o_m_m_e_n_t y−akses længde = 9 tommer;
xxx:= ymin;  yyy:= 0;
_f_o_r i:= 10 _s_t_e_p 10 _u_n_t_i_l 140 _d_o
_b_e_g_i_n plotline(ymin,i−10,ymin,i);  plotline(ymin,i,ymin+10×deltax,i);
  _i_f i=50 _t_h_e_n _b_e_g_i_n x:= ymin−36×deltax;
  x:= plotchar(5,x,50,0.175);  plotchar(16,x,50,0.175) _e_n_d;
  _i_f i=100 _t_h_e_n
  _b_e_g_i_n x:= ymin−54×deltax;  x:= plotchar(1,x,100,0.175);
  x:= plotchar(16,x,100,0.175);  plotchar(16,x,100,0.175);
  _e_n_d i=100;
_e_n_d i;
_f_o_r i:= −10 _s_t_e_p −10 _u_n_t_i_l −40 _d_o _b_e_g_i_n
  plotline(ymin,i+10,ymin,i);  plotline(ymin,i,ymin+10×deltax,i) _e_n_d;
plotline(ymin,0,entier(ymin+1),0);
plotline(entier(ymin+1),0,entier(ymin+1),−2);  i:= 1;
_f_o_r i:=i+1 _w_h_i_l_e entier(ymin+i)_<ymax _d_o
_b_e_g_i_n y:= entier(ymin+i);  plotline(y−1,0,y,0);  plotline(y,0,y,−2) _e_n_d;
plotline(entier(ymin+i−1),0,ymax,0);  plotline(ymax,−40,ymax−10×deltax,−40);
_f_o_r i:=−30 _s_t_e_p 10 _u_n_t_i_l 140 _d_o
_b_e_g_i_n plotline(ymax,i−10,ymax,i);  plotline(ymax,i,ymax−10×deltax,i) _e_n_d;
_f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l noofy _d_o plotline(ymin+(i−1)×dy,g[i−1],ymin+i×dy,g[i]);
i:=1;  lstep:= 0.1;
_f_o_r i:=i×10 _w_h_i_l_e entier(i×lamax)=0 _d_o lstep:= 1/i/10;
i:= 0;  y:= ymin;  lamda:= lamax;
_f_o_r i:= i+1 _w_h_i_l_e y<ymax _d_o
_b_e_g_i_n _f_o_r k:= 0 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
  _i_f abs(lamda−10|∧ (−k))<10−7 _t_h_e_n _b_e_g_i_n
    _i_f k=0 _t_h_e_n plotchar(1,ymin−8×deltax,−49,0.15)
    _e_l_s_e _b_e_g_i_n x:= y−k×8×deltax;  x:= plotchar(16,x,−49,0.03);
    _f_o_r i:=2 _s_t_e_p 1 _u_n_t_i_l k _d_o x:= plotchar(16,x,−49,0.15);



    plotchar(1,x,−49,0.15) _e_n_d
  _e_n_d;
  lamda:= lamda − lstep;
  _i_f lamda<10−7 _t_h_e_n _b_e_g_i_n lamda:= lamda+lstep−lstep/10;  lstep:= lstep/10 _e_n_d;
  y:= −ln(lamda);
  _i_f y>ymax _t_h_e_n _b_e_g_i_n plotline(−ln(lamda+lstep),−40,ymax,−40);  _g_o_t_o l1 _e_n_d;
  plotline(−ln(lamda+lstep),−40,y,−40);  plotline(y,−40,y,−42);
l1:  _e_n_d lamdaakse 1;
i:= 1;  lstep:= 0.1;  _f_o_r i:=i×10 _w_h_i_l_e entier(i×lamax)=0 _d_o lstep:=1/i/10;
i:= 0;  y:= ymin;  lamda:= lamax;
_f_o_r i:=i+1 _w_h_i_l_e y<ymax _d_o
_b_e_g_i_n lamda:= lamda − lstep;  _i_f lamda<10−7 _t_h_e_n
  _b_e_g_i_n lamda:= lamda+lstep−lstep/10;  lstep:= lstep/10 _e_n_d;  y:= −ln(lamda);
  _i_f y>ymax _t_h_e_n _b_e_g_i_n plotline(−ln(lamda+lstep),140,ymax,140);  _g_o_t_o l2 _e_n_d;
  plotline(−ln(lamda+lstep),140,y,140);  plotline(y,140,y,138);
l2:  _e_n_d lamdaakse 2;
_e_n_d næstyderste blok;
nykurve:= _f_a_l_s_e;  _g_o_t_o DATA;
_e_n_d yderste blok
_e_n_d som skal svare til plotprograms begin;


